
Wczytywanie i zapisywanie plików tekstowych przy pomocy PROC IMPORT

PROC IMPORT 
    DATAFILE="filename" | TABLE="tablename" 
    OUT=<libref.>SASdataset < (SASdatasetoptions)> 
    <DBMS=identifier>< REPLACE> ; 
  <datasourcestatement(s);> 

Przykład

PROC IMPORT DATAFILE="c:\sas\ego.csv" OUT=jeeshim.egov DBMS=CSV REPLACE;
   GETNAMES=YES;
   DATAROW=2;
RUN;

• DATAFILE określa plik do zaimportowania 
• OUT  zbiór SAS, który zostanie utworzony
• DBMS   typ importowanego pliku. Np.:,

• "DBMS=DLM" – plik tekstowy z separatorami (separator podaje się w instrukcji 
DELIMITER bloku PROC IMPORT, np.: DELIMITER=’;’  lub DELIMITER='09'x; ) 

• "DBMS=CSV" – separatorem są przecinki
• "DBMS=TAB" – separatorem są tabulacje

• REPLACE – nadpisanie istniejącego pliku
• GETNAMES – wczytuje nazwy zmiennych z pierwszej linii pliku
• DATAROW – określa pierwszy wiersz, który zostanie wczytany przez SASa

Przykład poniżej eksportuje zbiór danych do pliku tekstowego z polami oddzielonymi spacją. The DATA i 
OUTFILE oznaczają, odpowiednio, zbiór danych do wyeksportowania i zewnętrzny plik, do którego dane 
zostaną wyeksportowane.

PROC EXPORT
   DATA=jeeshim.egov
   OUTFILE="c:\sas\ego.txt"
   DBMS=DLM REPLACE;
RUN;

Wczytywanie i zapisywanie plików EXCEL przy pomocy PROC IMPORT

PROC IMPORT DATAFILE="c:\sas\assets.xls" OUT=js.pc DBMS=EXCEL2000 REPLACE;
   SHEET="computer";
   GETNAMES=YES;
RUN;

• instrukcja SHEET  określa arkusz do zaimportowania. Domyślnie jest to "sheet1", "sheet2", itd. 
• EXCEL2000 – oznacza format excel'a. Inne formaty:  EXCEL4, EXCEL5, and EXCEL97. 
• GETNAMES=YES – wczytuje nazwy zmiennych z pierwszego wiersza arkusza.

Poniższy przykład eksportuje zbiór danych do pliku excelowego..

PROC EXPORT  DATA=js.egov OUTFILE="c:\sas\egov.xls" DBMS=EXCEL2000 REPLACE;
RUN;

Instrukcja STOP

Instrukcja STOP kończy przetwarzanie bloku data. Przetwarzana jest pierwsza instrukcja po bloku DATA.

http://www.masil.org/sas/import.html#Top


Instrukcja SET

SET< SASdataset(s) <(datasetoptions(s) )>>    <options>; 

Bez argumentów SET wczytuje obserwację z ostatnio utworzonego zbioru danych. 

Argumenty

SASdataset 
    nazwa zbioru danych SAS

(datasetoptions)      
 Opcje określające akcje podejmowane przez SAS'a podczas odczytu zmiennych lub obserwacji do PDV.

DROP=
         wyłącza zmienne z przetwarzania 

data plan1 plan2; 
   set payroll(drop=salary gender); 
   if hired<’01jan98’d then output plan1; 
   else output plan2; 
run; 

FIRSTOBS=
Określa, którą obserwację SAS przetwarza jako pierwszą. W przykładzie użyto FIRSTOBS=  i 
OBS= aby odczytać obserwacje od 5 do 10 ze zbioru STUDY.
data new; 
   set study(firstobs=5 obs=10); 
run; 

IN= 
Tworzy zmienną która wskazuje czy zbiór danych jest źródłem danych dla aktualnej obserwacji. 
W przykładzie IN= ustala wartość 'i' na 1 gdy obserwacja została wczytana ze zbioru albo 0 w 
przeciwnym przypadku.  Instrukcja IFTHEN sprawdza wartość 'i' aby sprawdzić czy NONUSA 
jest źródłem danych aktualnej obserwacji.
   data allflts; 
      set usa nonusa(in=i); 
      by fltnum; 
      if i then overseas=’*’; 
   run; 

   data all2;          
     set males(in=m) females(in=f); 
     if m then sex=‘M’; 
     if f then sex=‘F’; 
   run; 

KEEP= 
Określa zmienne do przetwarzania.

OBS=
Określa ostatnią obserwację, która zostanie przetworzona. 



RENAME=
        Zmienia nazwę zmiennej

RENAME=(oldname1=newname1 < ...oldnamen=newnamen>)

W poniższym przykładzie RENAME= zostało użyte aby pokazać, że nazwa zmiennej jest 
zmieniana w momencie zapisywania do zbioru wynikowego. W czasie przetwarzania bloku 
DATA zmienna zachowuje oryginalną nazwę.
  data two(rename=(x=keys)); 
      set one; 
      z=x+y; 
  run; 

W tym przykładzie zmienna ma nową nazwę KEYS w całym bloku DATA. 
   data three; 
      set one(rename=(x=keys)); 
      z=keys+y; 
   run; 

WHERE= Data Set Option 
Wybiera obserwacje spełniające określone kryteria.
   data whizmo; 
      set sales(where=(product=’whizmo’)); 
   run; 

Options

KEY=index</UNIQUE> 
Umożliwia niesekwencyjny dostęp do obserwacji w zbiorze SAS bazujący na wartości zmiennej 
indeksowej lub klucza. W tym celu należy podać nazwę prostego lub złożonego klucza w zbiorze 
danych, który jest odczytywany. KEY= nie może być używany razem z  POINT= (patrz: Indeksy)

UNIQUE 
Sprawia, że KEY= szuka zawsze od początku indeksu w odczytywanym zbiorze danych. 
Domyślnie SET rozpoczyna szukanie od początku indeksu gdy wartość KEY ulega zmianie. Jeśli 
wartość KEY= nie zmienia się w kolejnych wykonaniach instrukcji SET przeszukiwanie 
rozpoczyna się od obserwacji następnej po ostatnio odczytanej. Ma to znaczenie przy odczycie 
obserwacji ze zduplikowanymi wartościami KEY=.

END=variable 
Tworzy i nazywa tymczasową zmienną będącą znacznikiem końca pliku. Zmienna przyjmująca 
początkową wartość 0, jest ustawiana na 1 gdy SET odczytuje ostatnią obserwację ostatniego 
wymienionego zbioru danych. Ta zmienna nie jest dodawana do zbioru wyjściowego. END= nie 
może być używana z POINT=. 

NOBS=variable 
Tworzy i nazywa tymczasową zmienną, której wartością jest zazwyczaj całkowita liczba 
obserwacji w zbiorze (lub zbiorach) wejściowych. Liczba obserwacji zawiera również te 
zaznaczone do usunięcia, lecz jeszcze nie usunięte. Deskryptor zbioru danych jest wczytywany w 
trakcie kompilacji bloku DATA jeszcze przed wykonaniem bloku. Na tej podstawie jest 
inicjowana zmienna przypisana przez NOBS=. Dzięki temu można odwoływać się do tej 
zmiennej jeszcze przed instrukcją SET. Zmienna nie jest dodawana do zbioru wyjściowego.

POINT=variable 
Tworzy i nazywa tymczasową zmienną, której wartość determinuje numer odczytywanej 
obserwacji. POINT= umożliwia bezpośredni (random) dostęp do zbioru danych.  Nie można 
używać POINT= z instrukcją BY, WHERE lub  opcją WHERE= instrukcji SET.  Nie można też 



użyć POINT= i  KEY= jednocześnie. Można użyć zmiennej określonej przez POINT= w pętli 
aby kolejno odczytywać obserwacje jakiegoś zbioru.

  
Opis

Za każdym razem gdy jest wykonywana instrukcja SET SAS wczytuje jedną obserwację do PDV. 
Instrukcja SET może dotyczyć wielu zbiorów danych, a blok DATA może zawierać wiele instrukcji SET.
Instrukcja SET ma wiele zastosowań w języku SAS. Są one określone przez opcje i instrukcje używane 
razem z instrukcją SAS i dotyczą: odczytu obserwacji i zmiennych z istniejących zbiorów danych do 
dalszej obróbki w bloku DATA,  dołączania zbiorów i odczytywania zbiorów w trybie jeden do jednego, 
odczytywania zbiorów w trybie dostępu bezpośredniego (random access).

BY – przetwarzanie grup z instrukcją SET 

Tylko jedna instrukcja BY może towarzyszyć instrukcji SET  w bloku DATA. Instrukcja BY powinna 
następować bezpośrednio po SET, której dotyczy. Zbiory danych podane w instrukcji SET muszą być 
posortowane po wartościach zmiennych, które są podane w instrukcji BY lub muszą mieć indeks. 
Instrukcja SET użyta z instrukcją BY pozwala na przeplatanie zbiorów danych będących argumentem 
instrukcji SET. Obserwacje  w zbiorze wynikowym są uporządkowane wg wartości zmiennych BY a 
wewnątrz grupy BY, w kolejności zbiorów danych, w których występują.

Łączenie zbiorów danych 

Użycie pojedynczej instrukcji SET z wieloma zbiorami danych pozwala połączyć (dołączyć jeden do 
drugiego) wyszczególnione zbiory. Tzn. liczba obserwacji w zbiorze wynikowym jest sumą obserwacji 
we wszystkich zbiorach wejściowych i na początku są obserwacje z pierwszego zbioru, potem z drugiego 
itd. Użycie wielu instrukcji SET z odpowiednimi parametrami pozwala wykonać złączenie jeden do 
jednego. 

Przykłady

Przykład 6: Łączenie jednej obserwacji z wieloma obserwacjami
Można odczytać obserwacje z jednego zbioru i połączyć je z obserwacjami z innego używając metod 
dostępu bezpośrednego: 
  data south; 
      set revenue; 
      if region=4; 
      set expense point=_n_; 
  run; 

Przykład 7:  Table Lookup     
Ten przykład pokazuje użycie opcji KEY= w celu odczytania danych z tabeli powiązanej na podstawie 
pewnej wartości w tabeli głównej. Blok  DATA wczytuje główny zbiór danych nazwany INVTORY i dane 
ze zbioru powiązanego PARTCODE. Używa przy tym indeksu PARTNO w celu niesekwencyjnego 
odczytania zbioru PARTCODE – na podstawie wartości PARTNO ze zbioru INVTORY wyszukuje w 
PARTCODE obserwację o tej samej wartości PARTNO. Celem jest uzyskanie odpowiedniego opisu, 
który jest zawarty tylko w zmiennej DECS w tabeli powiązanej.
   data combine; 
       set invtory(keep=partno instock price); 
       set partcode(keep=partno desc) key=partno; 
   run; 



Przykład 9: Odczyt obserwacji w trybie bezpośredniego adresowania
Poniższe instrukcje wybierają 50 obserwacji ze zbioru DRUGTEST przez użycie opcji POINT=: 
   data sample; 
      do obsnum=1 to 100 by 2; 
         set drugtest point=obsnum; 
         if _error_ then abort; 
         output; 
      end; 
      stop; 
   run; 

Przykład 10: Iteracja po wszystkich obserwacjach
Poniższe instrukcje używają NOBS= w celu ustalenia znacznika końca pętli. Wartość zmiennej 
tymczasowej LAST jest sumą liczby obserwacji w SURVEY1 i  SURVEY2: 
   do obsnum=1 to last by 100; 
       set survey1 survey2 point=obsnum nobs=last; 
               output; 
    end; 
    stop; 

Przykład 11: Zapisanie obserwacji dopiero po wczytaniu wszystkich obserwacji wejściowych
W tym przykładzie zmienna LAST w opcji END= otrzymuje wartość 1 tylko, gdy została wczytana 
ostatnia obserwacja:
     set rental end=last; 
     totdays + days; 
     if last then do; 
         revenue=totdays*65.78; 
         output; 
      end; 

Łączenie zbiorów danych SAS

Doczepianie
DATA newdataset; 
   SET dataset1 datasetn; 

Przeplatanie
DATA newdataset; 
   SET dataset1 datasetn; 
   BY variablelist; 

Łączenie jeden do jednego 
DATA newdataset; 
   MERGE dataset1 dataset2; 
(pierwsza obserwacja ze zbioru dataset1 połączona z pierwszą obserwacją ze zbioru dataset2, 
druga obserwacja ze zbioru dataset1 połączona z drugą obserwacją ze zbioru dataset2, itd)
  

Łączenie przez dopasowanie zmiennej
(wcześniej zbiory danych muszą być posortowane po wspólnej zmiennej)
DATA newdataset; 
   MERGE dataset1 dataset2; 
   BY variablelist; 



Dołączenie sumy do oryginalnych danych

       DATA newdataset; 
            IF _N_ = 1 THEN SET summarydataset; 
            SET originaldataset; 

W powyższym przykładzie originaldataset jest zbiorem z wieloma obserwacjami (oryginalne dane)  a 
summarydataset jest zbiorem z pojedynczą obserwacją (np. zawierającym sumę). SAS odczytuje kolejne 
obserwacje ze zbioru originaldataset w kolejnych iteracjach bloku DATA. Natomiast summarydataset 
jest odczytywany tylko w pierwszej iteracji DATA (gdy _N_ = 1).  Wartości wczytane ze zbioru 
summarydataset są zachowane dla wszystkich nowo utworzonych obserwacji obserwacji w newdataset. 
Jest to możliwe ponieważ zmienne wczytane przez instrukcję SET są automatycznie zachowywane, co 
zazwyczaj nie jest zauważalne, gdyż zachowane wartości są nadpisywane w następnej iteracji.

Zmienne automatyczne

FIRST.variable i LAST.variable
Są dostępne tylko w wyjątkowych sytuacjach podczas używania instrukcji BY w bloku DATA. 
FIRST.variable otrzymuje wartość  1  gdy SAS przetwarza obserwację, w kŧórej po raz pierwszy 
wystąpiła nowa wartość zmiennej 'variable' i 0 w pozostałych obserwacjach. Zmienna LAST.variable 
otrzymuje wartość 1 dla ostatniej obserwacji w której zmienna zachowuje poprzednią wartość (ostatniej 
obserwacji w grupie).

Indeksy

Nie zawsze są potrzebne – czasami wystarczy, że zbiór jest posortowany. Tworzenie przez opcję bloku 
DATA:

          DATA datasetname(INDEX=(varlist / <UNIQUE><NOMISS>  )); 

datasetname – nazwa nowego zbioru danych SAS.  Varlist – nazwa zmiennej która będzie kluczem lub 
lista zmiennych jeśli to będzie klucz złożony. Np.:

                  data cdsales.bighits(index=(cdnumber / unique)); 
                      set olddata.oldhits; 
                      ... more SAS Statements... 
                  run; 

Indeks może zostać wykorzystany w opcji KEY instrukcji SET. Opcja ta pozwala wczytać tylko wybrane 
obserwacje:
        SET datasetname KEY=indexname; 

Przykład:
         data cdsales; 
         set trans.hits(keep=cdnumber trnartist trnsales); 
         set olddata.oldhits key=cdnumber; 
                 ...more SAS statements... 
         run; 



Makrozmienne

Przechowują jedynie teksty. Ich wartości są ustalane przed kompilacją programu. Wartości są rozwijane 
także w cudzysłowach ale tylko w ” ”, a nie w ' '. 

Przykład:
%LET swieta=’24dec2002'd;
%LET zbior=klasa;

TITLE ”Raport dla zbioru &zbior”;
Ile_dni = &swieta – today();
put ‘Do Swiat zostało ‘ ile_dni ‘ dni;

%PUT – zapisuje makrozmienne do logu w czasie kompilacji
Np.
%PUT ***** &SYSDATE *****;

Makroprogramy

W makroprogramach możemy używać wyrażeń i funkcji wykonywanych przez makroporcesor, np.: %if, 
%then, %else, %do, %to, %end, %eval(), %syseval()

Przykład:
%MACRO CHART(NAME,BARVAR);
  PROC CHART DATA=&NAME;
    VBAR &BARVAR;
  RUN;
%MEND;
%CHART(PAYROLL,EMP)

Kompilacja warunkowa
%IF może warunkowo przekazywać kod do kompilatora.
Przykład: uruchomić PROC PRINT tylko jeśli PRTCH=YES.

%MACRO PTCHT(PRTCH,NAME,BARVAR);
    %IF &PRTCH=YES %THEN PROC PRINT DATA=&NAME;
        ;
     PROC CHART DATA=&NAME;
          VBAR &BARVAR;
     RUN;
%MEND;
%PTCHT(YES,PAYROLL,EMP)

Odczytanie liczby obserwacji w zbiorze

data _null_; 
   set demolib.stawki nobs=lstawek; 
   call symput('lstawek_mz', lstawek ); 
   stop; 
run; 

Makrozmienna lstawek_mz zawiera wartość zmiennej lstawek i może zostać użyta jak każda 
makrozmienna, czyli przez poprzedzenie znakiem &.


